
      ⍝   ⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝ ⊆ 
       
      ⍝   Monadic form: ⊆ ⍵ 
       
      ⍝   Frequently occurring pattern: enclose-if-simple 
       
      process 'red' 'green' 'blue'      ⍝ process several items 
       
      process 'purple'                  ⍝ or single item 
       
    ∇ process items;item  
[1]                       
[2]   :If 1=≡items        
[3]       items←⊂items    
[4]   :EndIf              
[5]                       
[6]   :For item :In items 
[7]        ⍝⍝⍝ do stuff   
[8]   :EndFor             
    ∇                     
 
becomes: 
 
    ∇ process items;item  
[1]   :For item :In ⊆items 
[2]        ⍝⍝⍝ do stuff   
[3]   :EndFor             
    ∇                     
       
      ⍝   definition: ⊆ ←→ {⊂⍣(1=≡⍵) ⊢⍵} 
       
      ⍝   Ex: think of an example from your own work. 
       
      ⍝   Dyadic form: ⍺ ⊆ ⍵  ⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝ 
       
      ⍝   definition: ⊆ ←→ {⎕ml←3 ⋄ ⍺⊂⍵}  ⍝ APL2-style partition 
      ⍝                    (may be applied under axis) 
       
      ⍝   ⊂ Dyalog default (⎕ml←1) partitioned enclose:       
      ⍝       left argument must be boolean vector or scalar 
      ⍝       new partition at _each_ 1       
      ⍝       leading 0s omit items 
       
      1 0 1 0 ⊂ 3 4 5 6 
┌───┬───┐ 
│3 4│5 6│ 
└───┴───┘ 
    



      0 1 0 1 ⊂ 3 4 5 6 
┌───┬─┐ 
│4 5│6│ 
└───┴─┘ 
      text 
some of your purple berries 
 
      ⍝   ⊆ Partition: 
      ⍝       left argument must be non-negative integer vector 
      ⍝       new partition when item > previous one (1,2</⍺) 
      ⍝       all 0s omit items 
 
      1 1 2 2 ⊆ 3 4 5 6 
┌───┬───┐ 
│3 4│5 6│ 
└───┴───┘ 
       
      1 0 2 2 ⊆ 3 4 5 6         
┌─┬───┐ 
│3│5 6│ 
└─┴───┘ 
       
      split←{(⍺≠⍵)⊆⍵}     ⍝ ⍺-split of ⍵ 
       
      ' ' split text 
┌────┬──┬────┬──────┬───────┐ 
│some│of│your│purple│berries│ 
└────┴──┴────┴──────┴───────┘ 
       
      split ← ≠⊆⊢         ⍝ coded as a fork 
       
      ' ' split text 
┌────┬──┬────┬──────┬───────┐ 
│some│of│your│purple│berries│ 
└────┴──┴────┴──────┴───────┘ 
       
      ⍝   Ex: Write ⊂ in terms of ⊆ 
      ⍝   Ex: Discover something interesting to share 
 
       
      ⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝⍝ @ 
       
      ⍝   {val} (mod @ sel) array 
 
      ⍝   Sub-array selection: 
      ⍝       numeric indexing 
      ⍝       boolean mask 
      ⍝       ... 
       
  



      ⍝   Right operand: [sel]ection: 
       
      ⍝       array: 
      ⍝           simple : major cell selection 
      ⍝           nested : choose / reach 
       
      ⍝       function: returns boolean mask 
 
      ⍝   Items of [array] are replaced with {val} mod sub, 
      ⍝   where [sub] is the sub-array of [sel]ected items. 
 
      ⍝   One last thing: this commonly occurring form: 
      ⍝       val (⊣ @ sel) array     
      ⍝   can be shortened to: 
      ⍝       (val @ sel) array 
 
      ⍝   Ex: write expressions for: 
      ⍝       vector ⍵ with 0s at alternate positions 
      ⍝       Sentence ⍵ with first letter of each word capital 
      ⍝       Exchanging rows p and q in matrix ⍵ 
      ⍝       FizzBuzz 
      ⍝   Ex: Discover something interesting to share 
       
       
      ⍝ Implementation: 
      ⍝ 0. Monitor usage in "real" apps. 
 
      ⍝ Performance improvements: 
      ⍝ Phase 1 (done): split single MAGIC fn into sub-cases so 
      ⍝  that case-selection occurs in C rather than APL. 
      ⍝ Phase 2 (done): process _function_ right operand in C- 
      ⍝   code to produce index array. 
      ⍝ Phase 3: Recode popular sub-cases into C to avoid inter- 
      ⍝ pretative overhead, and Unsharing() where possible. 
      ⍝ Phase 4: Further reduce Unsharing() by overwriting 
      ⍝ "garbage" right argument in-situ. 
      ⍝ Identifying the right argument as garbage might require 
      ⍝   some help from the compiler. 
 
      ⍝ In-situ substitution: 
      ⍝   0@1⍳⍵   ⍝ OK, because ⍳⍵ is not shared       
      ⍝   0@1⊢⍵   ⍝ OK, iff ⍵ has low refcount 
      ⍝ ⍵+0@1⊢⍵   ⍝ NOT OK 
      ⍝ v←0@1⊢⍵   ⍝ NOT OK if ⍵ is referenced downstream. 
      ⍝ 1+0@1⊢⍵   ⍝ NOT OK if a local error-guard references ⍵ 
      ⍝ ... etc 


